PROGRAM DESCRIPTION I

Program Title PROGRAMMED EQUATIONS

Contributor's Name Robert K. Meyer

Address Department of Electrical Engineering, University of Wyoming

City Laramie **State/Country** Wyoming **Zip Code** 82071

Program Description, Equations, Variables PREQ solves a system of N nonlinear equations for N=1 up to about N=12 in the form of

$$f_1(x_1,x_2,\ldots,x_N) = 0$$
$$f_2(x_1,x_2,\ldots,x_N) = 0$$
$$\vdots \qquad\qquad \vdots$$
$$f_N(x_1,x_2,\ldots,x_N) = 0$$

or in matrix form

$$\underline{F}(\underline{X}) = \underline{0}$$

The functions are defined by alphanumeric labels, i.e., LBL "1", $f_1(\underline{X})$, RTN, and use x_i found in register R(i), i.e., R01 contains x_1. Stand alone user subroutines include NR for single iteration of nonlinear equations using Newton-Raphson's method, JM to generate the Jacobian matrix and LE to solve a system of N linear equations by Gaussian elimination using complete pivoting to minimize roundoff errors. Time required per iteration is roughly $(6N)^2$ seconds. PREQ requires SIZE $(N+1)^2$.

Necessary Accessories One or two memory modules, printer optional

Operating Limits and Warnings Functions may use subroutines two deep. Newton-Raphson's method requires the initial guess for $\underline{X}$ "near" the solution to insure convergence.

Reference(s) Carnahan, Luther, Wilkes, Applied Numerical Methods, Wiley, N.Y., 1969

PROGRAM DESCRIPTION I

(CONTINUATION PAGE)

Description of alpha labels

PREQ PRogrammed EQuations prompts for data input, executes the iteration process, monitors progress and outputs results.

NR Performs a single iteration on a system of nonlinear equations by the Newton-Raphson method calling on JM and LE to give the (I+1)th approximation of the solution to $\underline{F}(\underline{X}) = \underline{0}$ where

$$\underline{X}_{I+1} = \underline{X}_I + \underline{\delta}_I$$

and $\underline{\delta}_I$ is the solution to the linear equation

$$\underline{\Phi}(\underline{X}_I)\underline{\delta}_I = -\underline{F}(\underline{X}_I)$$

JM Generates the Jacobian Matrix $\underline{\Phi}(\underline{X}_I)$ at the point $\underline{X}_I$ where

$$\underline{\Phi}(\underline{X}_I) = (f_{ij}(\underline{X}_I)) = f_i(\underline{X}_I)/\, x_j^{(I)},\ i=1,2,\ldots,N;\ j=1,2,\ldots,N$$

LE Solves a system of Linear Equations of the form

$$\underline{AX} = \underline{b}$$

NI Stores, increments and displays in the alpha register I, the Number of Iterations completed. Sets flag 02 if I=10.

CE Displays the Convergence Error E for the (I+1)th iteration where

$$E = \begin{cases} \dfrac{x_i^{(I+1)} - x_i^{(I)}}{x_i^{(I+1)}}, & x_i^{(I+1)} \neq 0 \\ x_i^{(I+1)} - x_i^{(I)}, & x_i^{(I+1)} = 0 \end{cases} \qquad i=1,2,\ldots,N$$

Flag 03 is set if $E \not\leq 1.00E-6$

OX Output X values via printer and/or alpha register and X register.

Subroutines called by two or more of the above routines

N Retrieves N from R00.

R Calculates the address of the m-th Row of the matrix where

$$R = (Nm + 1) + (Nm + N)/1000$$

where (Nm + 1) is the address of the first element in row m and (Nm + N) is the address of the last element in row m.

W Calculates the integer (N + C) from a number in the X register of the form C + N/1000.

X Converts an integer in the X register to an alpha number in the X register for indirect addressing of the functions.

Y Alpha register-stack manipulation.

Z Alpha register-stack manipulation.

PROGRAM DESCRIPTION I

(CONTINUATION PAGE)

Subroutine use and program modifications

NR The Newton-Raphson equation solver may be used to execute a single iteration as illustrated by example 2. In this case the alpha register is used only for indirect addressing and subroutine X may be shortened by deleting steps 29, 37 and 38. Subroutines Y and Z may then be omitted.

JM The Jacobian Matrix generator is illustrated by example 3. The modifications for NR are also applicable to this program.

LE The Linear Equation solver is illustrated by example 5. Three interchangeable versions are included, complete pivoting is recommended

Complete pivoting, slowest and most accurate. The first part of the program finds the largest unused pivot in the matrix and then performs the normalization and reduction process. The second part unscrambles the results by rearranging the rows of the system matrix into the unit matrix configuration.

Partial pivoting, next in length, faster and less accurate. The first part of the program finds the largest pivot beginning with the first column. The second part uses the "core" program to perform normalization and reduction.

No pivoting, fastest and least accurate. The first part of the program searches for any nonzero pivot beginning with the first column. The second part uses the "core" program to perform normalization and reduction. The "core" program, steps 32 through 92 in this version of LE could be used to solve linear equations; however, no provisions are made to handle a zero pivot. ISG 00 must be inserted immediately after LBL 01 in order to use the "core" program alone.

Run times for example 5 for complete pivoting, partial pivoting and no pivoting are 80, 50 and 35 seconds respectively. About 20 percent difference results in run times for PREQ whether the complete pivoting or no pivoting program is used.

NI The flag for IMAX value may be changed in PREQ, line 116.

CE Convergence test magnitude may be changed in PREQ, line 161.

OX Display format may be changed in PREQ, line 180. The remainder of the program reverts to FIX 4 format except during alpha register use.

PROGRAM DESCRIPTION I

(CONTINUATION PAGE)

Subroutine requirements



PROGRAM DESCRIPTION II

Sample Problems — Nonlinear Examples

$f_1(x_1,x_2) = (\sin(x_1x_2) - x_1)/2 - x_2/(4\pi) = 0$

$f_2(x_1,x_2) = (x_2/\pi - 2x_1 - (\exp(2x_1 - 1) - 1)(1/(4\pi) - 1))\exp(1) = 0$

Initial guesses for $\underline{F}(\underline{X})=\underline{0}$ and point for $\underline{\mathfrak{D}}(X_I)$ evaluation are x_1=0.4 and x_2=3.0

1. Solve equations using PREQ
2. Solve equations using NR
3. Generate the Jacobian matrix using JM

```
01◆LBL "1"
02 RCL 01
03 RCL 02
04 *
05 RAD
06 SIN
07 DEG
08 RCL 01
09 -
10 2
11 /
12 RCL 02
13 4
14 /
15 PI
16 /
17 -
18 RTN
19◆LBL "2"
20 RCL 02
21 PI
22 /
23 2
24 RCL 01
25 *
26 -
27 LASTX
28 1
29 -
30 E↑X
31 1
32 -
33 4
34 PI
35 *
36 1/X
37 1
38 -
39 *
40 -
41 1
42 E↑X
43 *
44 RTN
```

```
EXAMPLE 1
                XEQ "PREQ"
N=?
              2.0000    RUN
X1=?
               .4000    RUN
X2=?
              3.0000    RUN
READY
                        RUN

X1=4.000E-1
X2=3.000E0
I: 0

X1=-4.309E-1
X2=1.751E0
I: 1, E:1.93E0

X1=-2.455E-1
X2=7.331E-1
I: 2, E:1.39E0

X1=-2.614E-1
X2=6.189E-1
I: 3, E:1.85E-1

X1=-2.606E-1
X2=6.225E-1
I: 4, E:5.81E-3

X1=-2.606E-1
X2=6.225E-1
I: 5, E:9.08E-6

X1=-2.606E-1
X2=6.225E-1
I: 6, E:3.84E-10
                  "FIN"

                XEQ "1"
      3.0000-11    ***
                XEQ "2"
     2.7183-10     ***
```

```
EXAMPLE 1 CONTINUED
                XEQ "QX"

X1=-2.606E-1
X2=6.225E-1
------------
EXAMPLE 2
          .0020 STO 00
          .4000 STO 01
         3.0000 STO 02
                XEQ "NR"

R01= -0.4309
R02=  1.7511
                XEQ "NR"

R01= -0.2455
R02=  0.7331
                XEQ "NR"

R01= -0.2614
R02=  0.6189
                XEQ "NR"

R01= -0.2606
R02=  0.6225
                XEQ "NR"

R01= -0.2606
R02=  0.6225
CONVERGED
------------
EXAMPLE 3
          .0040 STO 00
          .4000 STO 01
         3.0000 STO 02
                XEQ "JM"

R01=  0.4000   x1
R02=  3.0000   x2
R03=  0.0435   f11
R04= -0.0071   f12
R05= -1.3396   f21
R06=  0.8653   f22
```

PROGRAM DESCRIPTION II

Sample Problems

Linear Examples

$$-3x_1 + 8x_2 + 5x_3 = 6$$
$$x_1 + 9x_2 - 6x_3 = 1$$
$$2x_1 - 7x_2 + 4x_3 = 9$$

Note: To use PREQ, equations must be rewritten in the form $\underline{F}(\underline{X}) = 0$

Note: Using initial guesses of all zeroes or all ones makes $\underline{\Phi}(\underline{X}) = \underline{A}$ when using PREQ

4. Solve equations using PREQ
5. Solve equations using LE

```
01◆LBL "1"
02 -3
03 RCL 01
04 *
05 8
06 RCL 02
07 *
08 +
09 5
10 RCL 03
11 *
12 +
13 6
14 -
15 RTN
16◆LBL "2"
17 1
18 RCL 01
19 *
20 9
21 RCL 02
22 *
23 +
24 -6
25 RCL 03
26 *
27 +
28 1
29 -
30 RTN
31◆LBL "3"
32 2
33 RCL 01
34 *
35 -7
36 RCL 02
37 *
38 +
39 4
40 RCL 03
41 *
42 +
43 9
44 -
45 RTN
```

EXAMPLE 4

```
                XEQ "PREQ"
N=?
          3.0000    RUN
SET SIZE 16
                 SIZE 016
                     RUN
X1=?
          1.0000    RUN
X2=?
          1.0000    RUN
X3=?
          1.0000    RUN
READY
                     RUN

X1=1.000E0
X2=1.000E0
X3=1.000E0
I: 0

X1=4.000E0
X2=1.000E0
X3=2.000E0
I: 1, E:7.50E-1

X1=4.000E0
X2=1.000E0
X3=2.000E0
I: 2, E:0.00E0
                    "FIN"
                XEQ "0X"

X1=4.000E0
X2=1.000E0
X3=2.000E0
```

EXAMPLE 5

```
          .0030 STO 00
          6.0000 STO 01
          1.0000 STO 02
          9.0000 STO 03
         -3.0000 STO 04
          8.0000 STO 05
          5.0000 STO 06
          1.0000 STO 07
          9.0000 STO 08
         -6.0000 STO 09
          2.0000 STO 10
         -7.0000 STO 11
          4.0000 STO 12
              XEQ "LE"

R01=  4.0000
R02=  1.0000
R03=  2.0000
```

USER INSTRUCTIONS

PREQ PRogrammed EQuations — SIZE: (HP-41C) >1

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load PREQ, NR, JM, LE, N, R, W, X, Y, Z			
	Programs may be merged if desired			
2.	Prepare to load function $\underline{F}(\underline{X})$		GTO..	
3.	Switch to PRGM mode; load $f_i(\underline{X})$ under			
	LBL "i" using x_i in register R(i); add			
	RTN for i = 1,2,...,N			
4.	Initialize program		XEQ "PREQ"	N=?
5.	Key in order N	N	R/S	SET SIZE nnn
6.	Set size and continue		XEQ "SIZE" nnn	
			R/S	X1=?
7.	Input initial guess (zero permissible)	x_1	R/S	X2=?
8.	Repeat step 7 for all x's	x_N	R/S	READY
9.	Solve equations		R/S	I: 0
	TONE 9, display iteration completed			I:nn, E:nnn
	I and convergence error E			
	Messages after program stops:			
	BEEP, $E=10^{-6}$, iteration finished,			FIN
	solution found, go to step 10			
	TONE 0, $\underline{\Phi}(\underline{X}_I)$ is singular, go to			DET=0
	step 4 and try another guess			
	TONE 0, I=10, not converged, continue			IMAX
	or go to step 4, try another guess			
	All cases: Display I and E		ALPHA	I:nn, E:nnn
	Resume iteration		R/S	
10.	Output $\underline{X}$ values		XEQ "OX"	X1=
11.			R/S	X2=
12.	Repeat step 11 for all x's		R/S	ALL

USER INSTRUCTIONS

NR Newton-Raphson

SIZE: (HP-41C) >1

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load NR, JM, LE, N, R, W, X, Y, Z			
	Programs may be merged if desired			
2.	Prepare to load function $\underline{F}(\underline{X})$		GTO..	
3.	Switch to PRGM mode; load $f_i(\underline{X})$ under			
	LBL "i" using x_i in register R(i); add			
	RTN for i = 1,2,...,N			
4.	Initialize program counter	N/1000	STO 00	
5.	Set size nnn = $(N+1)^2$		XEQ "SIZE" nnn	
6.	Input initial guess	x_1	STO 01	
7.	Repeat step 6 for all x's	x_N	STO (N)	
8.	Execute one iteration		XEQ "NR"	
	If program stops with flag 01 set,			
	$\underline{\underline{\mathbf{M}}}(\underline{X}_I)$ is singular, go to step 6 and			
	try another guess			
9.	Output $\underline{X}$ values		RCL 01	x_1
10.	Repeat step 9 for all x's		RCL (N)	x_N
11.	Repeat step 8 for desired convergence			

USER INSTRUCTIONS

JM Jacobian Matrix

SIZE: (HP-41C) >1

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load JM, N, R, W, X, Y, Z			
	Programs may be merged if desired			
2.	Prepare to load function $\underline{F}(\underline{X})$		GTO..	
3.	Switch to PRGM mode; load $f_i(\underline{X})$ under			
	LBL "i" using x_i in register R(i); add			
	RTN for i = 1,2,...,N			
4.	Initialize program counter	$N^2/1000$	STO 00	
5.	Set size nnn = $N^2 + N + 1$		XEQ "SIZE" nnn	
6.	Input point $\underline{X}_I$	x_1	STO 01	
7.	Repeat step 6 for all x's	x_N	STO (N)	
8.	Generate the Jacobian matrix		XEQ "JM"	
9.	Output $\underline{\Phi}$ in order of f_{11}, f_{12}, f_{13},		RCL (N+1)	f_{11}
	...f_{21}, f_{22}, f_{23},..., f_{NN}		RCL (N+2)	f_{12}
			.	.
			.	.
			.	.
10.	Repeat step 9 for all f_{ij}'s		RCL ($N+N^2$)	f_{NN}

USER INSTRUCTIONS

LE Linear Equations

SIZE: (HP-41C) >1

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load LE, N, R			
	Programs may be merged if desired			
2.	Initialize program counter	N/1000	STO 00	
3.	Set size nnn = $N^2 + N + 1$		XEQ "SIZE" nnn	
4.	Input b	b_1	STO 01	
5.	Repeat step 4 for all b's	b_N	STO (N)	
6.	Input A in order of a_{11}, a_{12}, a_{13},	a_{11}	STO (N+1)	
	$\cdots, a_{21}$, a_{22}, $a_{23}, \cdots$, a_{NN}	a_{12}	STO (N+2)	
		$\vdots$	$\vdots$	
7.	Repeat step 6 for all a's	a_{NN}	STO $(N+N^2)$	
8.	Solve equations		XEQ "LE"	
	If program stops with flag 01 set,			
	A is singular			
9.	Output solution X		RCL 01	x_1
10.	Repeat step 9 for all x's		RCL (N)	x_N

012010

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

PREQ PRogrammed EQuations

STEP/ LINE	KEY ENTRY	COMMENTS
01	LBL "PREQ"	PRogrammed EQuations
02	"N=?"	
03	PROMPT	Input order
04	3	Initialize
05	10↑X	
06	/	
07	STO 00	
08	ISG 00	
09	CF 00	
10	LBL 15	Size check
11	1	
12	ENTER↑	
13	XEQ "N"	
14	+	
15	X↑2	
16	-	
17	FIX 0	
18	CF 29	
19	SF 25	
20	STO IND X	
21	"SET SIZE "	
22	ARCL L	
23	FC? 25	
24	PROMPT	
25	FC?C 25	
26	GTO 15	
27	"X"	
28	ARCL 00	
29	"⊢=?"	
30	FIX 4	
31	SF 29	
32	PROMPT	Input initial guess
33	STO IND 00	
34	ISG 00	
35	GTO 15	Next guess
36	"READY"	
37	PROMPT	All data in
38	FC? 55	Initialize display
39	CF 21	
40	FS? 55	
41	XEQ "OX"	Printer output
42	"I: 0"	
43	AVIEW	
44	LBL 16	Iteration routine
45	XEQ "NR"	
46	XEQ "NI"	
47	XEQ "CE"	
48	FS? 55	
49	XEQ "OX"	Printer output
50	AVIEW	Display progress
51	FS? 01	Set if DET=0
52	XEQ 21	
53	FS?C 01	
54	STOP	
55	FS? 03	
56	CF 02	
57	FS? 02	Set if I=IMAX
58	XEQ 22	
59	FS?C 02	
60	STOP	
61	FS? 03	Set if FINished
62	XEQ 23	
63	FS?C 03	
64	STOP	
65	TONE 9	
66	GTO 16	Next iteration
67	LBL "NI"	Number of Iterations
68	CF 02	
69	XEQ "Y"	Isolate I
70	"**"	
71	XEQ "Z"	
72	XEQ "Y"	
73	"**"	
74	ARCL Y	
75	ASHF	
76	ASTO Y	
77	RDN	
78	-1	Initial T
79	X<>Y	
80	R↑	
81	FIX 0	
82	CF 29	
83	LBL 17	Decode numerical value of I
84	RDN	
85	X<>Y	
86	9	
87	X<=Y?	Test if T is one or two digits
88	CLA	
89	X>Y?	
90	" "	
91	RDN	
92	X<>Y	

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C PREQ PRogrammed EQuations (Continuation)

STEP/ LINE	KEY ENTRY	COMMENTS
93	R↑	
94	ISG Z	Increment T
95	STO X	
96	ARCL Z	Convert T to
97	ASTO X	alpha number
98	X≠Y?	T≠I?
99	GTO 17	Next T
100	ISG Z	Increment T=I
101	CLA	Rebuild alpha
102	CLA	register
103	"I:"	
104	RDN	
105	RDN	
106	9	
107	X<>Y	
108	X<=Y?	
109	"⊢ "	
110	X<>Y	
111	RDN	
112	ENTER↑	
113	XEQ "Z"	
114	FIX 4	
115	SF 29	
116	10	
117	X≠Y?	I≠IMAX
118	RTN	
119	SF 02	I=IMAX
120	RTN	
121	*LBL "CE"	Convergence
122	CF 00	Error
123	0	
124	XEQ "R"	
125	STO 00	
126	SF 03	
127	XEQ 18	Calculate $E=E_1$
128	*LBL 18	Calculate E_i,
129	1	convergence error
130	RCL IND 00	for x_i
131	X≠0?	
132	*	
133	X=0?	
134	+	
135	XEQ "W"	
136	LASTX	
137	X↑2	
138	+	
139	RDN	
140	RCL IND T	
141	RCL IND 00	
142	-	
143	X<>Y	
144	/	
145	ABS	E_i calculated
146	FS?C 03	
147	RTN	
148	X>Y?	E_i>E?
149	X<>Y	E_i becomes E
150	RDN	
151	ISG 00	
152	GTO 18	Test next x_i
153	"⊢, "	Update alpha
154	ASTO Y	register with
155	CLA	new E
156	ARCL Y	
157	"⊢E:"	
158	SCI 2	
159	ARCL X	
160	FIX 4	
161	-6	
162	10↑X	
163	X<Y?	E≤1.00E-6?
164	RTN	
165	SF 03	Converged
166	RTN	
167	*LBL "OX"	Output X
168	RCL 00	Initialize
169	FRC	
170	ISG X	
171	STO 00	
172	ADV	
173	XEQ "Y"	
174	*LBL 19	Rebuild alpha
175	FIX 0	register
176	CF 29	
177	"X"	
178	ARCL 00	
179	"⊢="	
180	SCI 3	
181	SF 29	
182	RCL IND 00	x_i in X register
183	ARCL IND 00	x_i in alpha register

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

012010

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C PREQ PRogrammed EQuations (Continuation)

STEP/ LINE	KEY ENTRY	COMMENTS	STEP/ LINE	KEY ENTRY	COMMENTS
184	AVIEW				
185	FC? 55				
186	STOP				
187	RDN				
188	ISG 00				
189	GTO 19	Next x_i			
190	FIX 4				
191	"ALL"	Message: ALL			
192◆	LBL 20	Display message			
193	ASTO X				
194	CLA				
195	XEQ "Z"				
196	CLD				
197	RTN				
198◆	LBL 21	Message: DET=0			
199	XEQ "Y"				
200	"DET=0"				
201	XEQ 20				
202	TONE 0				
203	VIEW X				
204	RTN				
205◆	LBL 22	Message: IMAX			
206	XEQ "Y"				
207	"IMAX"				
208	XEQ 20				
209	TONE 0				
210	VIEW X				
211	RTN				
212◆	LBL 23	Message: FIN			
213	XEQ "Y"				
214	"FIN"				
215	XEQ 20				
216	BEEP				
217	VIEW X				
218	RTN				

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

NR Newton-Raphson

STEP/ LINE	KEY ENTRY	COMMENTS
01	◆LBL "NR"	Newton-Raphson
02	RCL 00	Initialize
03	FRC	
04	X↑2	
05	3	
06	10↑X	
07	*	
08	STO 00	
09	XEQ "JM"	Generate $\Phi(X_I)$
10	0	
11	XEQ "R"	
12	CF 00	
13	STO 00	
14	◆LBL 10	Calculate $-F(X)$
15	RCL 00	
16	XEQ "X"	
17	CHS	
18	XEQ "W"	
19	LASTX	
20	X↑2	
21	+	
22	X<>Y	
23	STO IND Y	
24	ISG 00	
25	GTO 10	Next element
26	0	
27	XEQ "R"	
28	STO 00	
29	◆LBL 11	Exchange X and $-F$
30	XEQ "W"	
31	LASTX	
32	X↑2	
33	+	
34	RCL IND 00	
35	X<> IND Y	
36	STO IND 00	
37	ISG 00	
38	GTO 11	Next element
39	RCL 00	
40	FRC	
41	STO 00	
42	XEQ "LE"	Calculate δ_I
43	FS? 01	DET=0?
44	RTN	
45	0	
46	XEQ "R"	
47	STO 00	
48	◆LBL 12	Calculate X_{I+1}
49	XEQ "W"	
50	LASTX	
51	X↑2	
52	+	
53	RCL IND X	
54	ST+ IND 00	
55	ISG 00	
56	GTO 12	Next element
57	RTN	Completed

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

JM Jacobian Matrix

STEP/ LINE	KEY ENTRY	COMMENTS
01	*LBL "JM"	Jacobian Matrix
02	SF 00	Initialize
03	ISG 00	Increment ij
04	GTO 13	
05	RTN	
06	*LBL 13	
07	XEQ 13	
08	RCL IND Y	
09	XEQ 14	
10	LASTX	
11	+	
12	X<> IND Z	x_j+dx_j
13	XEQ "W"	
14	X<>Y	
15	STO IND Y	
16	RCL Z	
17	XEQ "X"	$f_i(x_j+dx_j)$
18	XEQ "W"	
19	X<>Y	
20	X<> IND Y	
21	XEQ 13	
22	R↑	
23	STO IND Z	x_j
24	X<>Y	
25	XEQ "X"	$f_i(x_j)$
26	XEQ "W"	
27	X<>Y	
28	ST- IND Y	
29	RDN	
30	XEQ 13	
31	RDN	
32	RCL IND X	
33	XEQ 14	
34	ST/ IND Z	$R(N+C)$: $f_{ij}=\frac{f_i(x_j+dx_j)-f_i(x_j)}{dx_j}$
35	GTO "JM"	
36	*LBL 13	Calculate i and j
37	XEQ "N"	
38	RCL 00	ROO: $C+N^2/1000$
39	INT	
40	RCL Y	
41	MOD	
42	X=0?	
43	X<>Y	j=N, N MOD C = 0
44	X<>Y	j=N MOD C,
45	RDN	N MOD C ≠ 0
46	LASTX	
47	RCL 00	
48	DSE X	
49	INT	
50	X<>Y	
51	/	
52	INT	
53	1	
54	+	$i=1+\text{Integer}\frac{(C-1)}{N}$
55	RTN	
56	*LBL 14	Calculate dx
57	-5	
58	10↑X	
59	X<>Y	
60	X≠0?	$dx=10^{-5}x$, $x\neq 0$
61	*	
62	X=0?	$dx=10^{-5}$, $x=0$
63	+	
64	RTN	

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

LE Linear Equations
Complete Pivoting

STEP/ LINE	KEY ENTRY	COMMENTS
01	LBL "LE"	Linear Equations
02	CF 00	Complete pivoting
03	CF 01	Initialize
04	ISG 00	Increment primary
05	RCL 00	counter k
06	XEQ "R"	
07	INT	
08	XEQ "N"	
09	XEQ "R"	
10	FRC	
11	+	
12	ENTER↑	
13	LBL 01	Obtain address of
14	RCL IND Y	largest element, k-th row and
15	ABS	below
16	RCL IND Y	
17	ABS	
18	X>Y?	Test new element
19	GTO 03	New element is
20	RDN	larger than
21	RDN	previous element
22	ISG X	
23	GTO 01	Test next element
24	RDN	Pivot is found
25	RCL 00	
26	LBL 02	Find row
27	STO Z	contianing pivot
28	XEQ "R"	
29	X>Y?	
30	GTO 00	
31	X<> Z	
32	1	
33	+	
34	GTO 02	
35	LBL 03	Replace previous
36	RDN	element with
37	RDN	larger element
38	ENTER↑	
39	GTO 01	Test next element
40	LBL 00	Find column
41	X<> Z	containing pivot
42	1	
43	-	
44	STO Z	
45	XEQ "R"	
46	-	
47	LASTX	
48	X<>Y	
49	INT	
50	X<>Y	
51	RCL IND 00	Exchange b of
52	X<> IND T	pivot row with b
53	STO IND 00	of k-th row
54	RDN	
55	RCL 00	
56	XEQ "R"	
57	LBL 04	Exchange a's of
58	RCL IND X	pivot row with
59	X<> IND Z	a's of k-th row
60	STO IND Y	
61	RDN	
62	1	
63	+	
64	ISG Y	
65	GTO 04	
66	RCL Z	Exchange complete
67	R↑	
68	RCL 00	
69	XEQ "R"	
70	+	
71	LASTX	
72	RCL IND Y	Recall pivot
73	X=0?	Singular if zero
74	SF 01	
75	X=0?	
76	RTN	Exit if singular
77	ST/ IND 00	Normalize b
78	LBL 05	Normalize a's
79	ST/ IND Y	
80	ISG Y	
81	GTO 05	
82	R↑	Normalization
83	0	completed
84	XEQ "R"	
85	LBL 06	Check so as not
86	XEQ "R"	to reduce pivot
87	RCL 00	row with itself
88	XEQ "R"	
89	X<>Y	
90	ST+ T	
91	RCL IND T	
92	STO L	

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

LE Linear Equations (Continuation)
Complete Pivoting

STEP/ LINE	KEY ENTRY	COMMENTS
93	RDN	
94	X≠Y?	Row counter≠k?
95	GTO 07	
96	FRC	Increment row
97	3	counter so as not
98	10↑X	to reduce pivot
99	*	row with itself
100	XEQ "N"	
101	/	
102	R↑	
103	X<>Y	
104	XEQ "N"	
105	X<Y?	N<row counter?
106	GTO 00	Solution completed
107	RDN	
108	GTO 06	Reduce next row
109	◆LBL 07	Reduce a's
110	RCL IND Y	
111	X<> L	
112	ST* L	
113	X<> L	
114	ST- IND Y	
115	RDN	
116	ISG X	
117	STO X	
118	ISG Y	
119	GTO 07	
120	X<>Y	Reduce b
121	RDN	
122	LASTX	
123	X<>Y	
124	FRC	
125	3	
126	10↑X	
127	*	
128	XEQ "N"	
129	/	
130	DSE X	
131	X<>Y	
132	RCL IND 00	
133	*	
134	ST- IND Y	Row is reduced
135	RDN	
136	1	
137	+	Increment row
138	XEQ "N"	counter
139	X<Y?	N-th row reduced?
140	GTO "LE"	Next pivot
141	RDN	
142	GTO 06	Reduce next row
143	◆LBL 00	Unscramble matrix
144	0	Initialize
145	XEQ "R"	
146	STO 00	
147	◆LBL 08	Search for row
148	RCL 00	that belongs in
149	XEQ "R"	the k-th row
150	RCL 00	
151	INT	
152	1	
153	-	
154	+	
155	RCL IND X	
156	X≠0?	Pivot nonzero?
157	GTO 00	Row OK
158	RCL IND 00	Exchange b to k-th row
159	X<> IND T	
160	STO IND 00	
161	R↑	
162	ENTER↑	
163	XEQ "R"	
164	RCL 00	
165	XEQ "R"	
166	◆LBL 09	Exchange a's to
167	RCL IND X	k-th row
168	X<> IND Z	
169	STO IND Y	
170	RDN	
171	1	
172	+	
173	ISG Y	
174	GTO 09	
175	RCL Z	Row is exchanged
176	1	
177	+	Increment row
178	GTO 08	Check next row
179	◆LBL 00	Increment k
180	RCL 00	
181	1	
182	+	
183	ISG 00	
184	GTO 08	Check next row
185	RTN	Matrix unscrambled

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

012010

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

N,R,W,X,Y,Z Subroutines

STEP/ LINE	KEY ENTRY	COMMENTS
01	LBL "N"	N subroutine
02	RCL 00	
03	FRC	
04	STO L	
05	RDN	
06	1 E3	
07	ST* L	
08	X<> L	
09	FS? 00	
10	SQRT	
11	RTN	
12	LBL "R"	R subroutine
13	INT	
14	XEQ "N"	
15	*	
16	ST+ L	
17	1 E3	
18	ST/ L	
19	X<> L	
20	+	
21	ISG X	
22	RTN	
23	LBL "W"	W subroutine
24	RCL 00	
25	XEQ "N"	
26	+	
27	RTN	
28	LBL "X"	X subroutine
29	XEQ "Y"	
30	CLA	
31	FIX 0	
32	CF 29	
33	ARCL X	
34	FIX 4	
35	SF 29	
36	ASTO X	
37	CLA	
38	XEQ "Z"	
39	XEQ IND X	
40	RTN	
41	LBL "Y"	Y subroutine
42	ASTO Y	
43	ASHF	
44	ASTO Z	
45	ASHF	
46	ASTO L	
47	RTN	
48	LBL "Z"	Z subroutine
49	ARCL Y	
50	ARCL Z	
51	ARCL L	
52	RTN	

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

LE Linear Equations
Partial Pivoting

STEP/ LINE	KEY ENTRY	COMMENTS
01	LBL "LE"	Linear Equations
02	LBL 01	Partial pivoting
03	CF 00	Initialize
04	CF 01	
05	ISG 00	Increment k
06	RCL 00	Begin with m=k
07	LBL 06	Obtain address of
08	ENTER↑	largest element
09	XEQ "R"	in column k
10	RCL 00	
11	XEQ "R"	
12	RCL 00	
13	INT	
14	DSE X	
15	STO X	
16	ST+ Z	
17	+	
18	R↑	
19	RCL IND Y	Recall a_{kk}
20	X<0?	
21	CHS	
22	RCL IND T	Recall a_{mk}
23	ABS	
24	X<=Y?	$\lvert a_{mk}\rvert \leq \lvert a_{kk}\rvert$?
25	GTO 01	Check next row
26	RCL IND 00	Exchange b of
27	X<> IND T	pivot row with b of m-th row
28	STO IND 00	
29	R↑	
30	ENTER↑	
31	XEQ "R"	
32	RCL 00	
33	XEQ "R"	
34	LBL 07	Exchange a's of
35	RCL IND X	pivot row with a's of m-th row
36	X<> IND Z	
37	STO IND Y	
38	RDN	
39	1	
40	+	
41	ISG Y	
42	GTO 07	
43	RCL Z	Exchange complete
44	GTO 06	Check next row
45	LBL 00	Matrix singular
46	SF 01	
47	RTN	Exit
48	LBL 01	
49	RCL Z	
50	ISG X	Increment m
51	GTO 06	Check next row
52	LBL 01	Core program
53	RCL 00	
54	XEQ "R"	Obtain address of
55	XEQ "N"	pivot for k-th
56	1	row
57	+	
58	RCL 00	
59	INT	
60	*	
61	RCL IND X	Recall pivot
62	X=0?	Pivot zero?
63	GTO 00	
64	LBL 02	Normalize a's
65	ST/ IND Z	
66	ISG Z	
67	GTO 02	
68	ST/ IND 00	Normalize b
69	0	
70	XEQ "R"	Normalization completed
71	LBL 03	Check so as not to
72	ENTER↑	reduce pivot row
73	XEQ "R"	with itself
74	RCL 00	
75	XEQ "R"	
76	RCL 00	
77	INT	
78	DSE X	
79	STO X	
80	ST+ Z	
81	+	
82	RCL IND Y	
83	STO L	
84	RDN	

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

LE Linear Equations (Continuation)
Partial Pivoting

STEP/ LINE	KEY ENTRY	COMMENTS
85	X≠Y?	Row counter ≠ k?
86	GTO 05	Reduce row
87	ISG Z	Increment row
88	GTO 04	counter
89	RTN	Solution completed
90	◆LBL 04	Recall row counter
91	RCL Z	
92	GTO 03	Next row
93	◆LBL 05	Reduce a's
94	RCL IND X	
95	X<> L	
96	ST* L	
97	X<> L	
98	ST- IND Z	
99	RDN	
100	ISG Y	
101	STO X	
102	ISG X	
103	GTO 05	
104	RCL Z	
105	RCL IND 00	Reduce b
106	LASTX	
107	*	
108	ST- IND Y	
109	RDN	
110	ISG X	
111	GTO 03	Reduce next row
112	GTO 01	Next pivot

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☒ 41C

LE Linear Equations
No Pivoting

STEP/ LINE	KEY ENTRY	COMMENTS
01	◆LBL "LE"	Linear Equations
02	◆LBL 01	No pivoting
03	CF 00	Initialize
04	CF 01	
05	ISG 00	Increment k
06	RCL 00	Begin with m=k
07	GTO 01	Go to Core prgm
08	◆LBL 00	Obtain pivot ≠ 0
09	RDN	
10	RDN	
11	ISG Y	Increment m
12	GTO 04	Exchange rows
13	SF 01	Matrix singular
14	RTN	Exit
15	◆LBL 04	Exchange b of
16	RCL IND 00	pivot row with b of m-th row
17	X<> IND Z	
18	STO IND 00	
19	RDN	
20	RCL Y	
21	XEQ "R"	
22	◆LBL 06	Exchange a's of
23	RCL IND X	pivot row with a's of m-th row
24	X<> IND Z	
25	STO IND Y	
26	RDN	
27	1	
28	+	
29	ISG Y	
30	GTO 06	
31	R↑	Exchange completed
32	◆LBL 01	Core program
33	RCL 00	
34	XEQ "R"	Obtain address of
35	XEQ "N"	pivot for k-th row
36	1	
37	+	
38	RCL 00	
39	INT	
40	*	
41	RCL IND X	Recall pivot
42	X=0?	Pivot zero?
43	GTO 00	
44	◆LBL 02	Normalize a's
45	ST/ IND Z	
46	ISG Z	
47	GTO 02	
48	ST/ IND 00	Normalize b
49	0	Normalization
50	XEQ "R"	completed
51	◆LBL 03	Check so as not to
52	ENTER↑	reduce pivot row
53	XEQ "R"	with itself
54	RCL 00	
55	XEQ "R"	
56	RCL 00	
57	INT	
58	DSE X	
59	STO X	
60	ST+ Z	
61	+	
62	RCL IND Y	
63	STO L	
64	RDN	
65	X≠Y?	Row counter ≠ k?
66	GTO 05	Reduce row
67	ISG Z	Increment row
68	GTO 04	counter
69	RTN	Solution completed
70	◆LBL 04	Recall row counter
71	RCL Z	
72	GTO 03	Next row
73	◆LBL 05	Reduce a's
74	RCL IND X	
75	X<> L	
76	ST* L	
77	X<> L	
78	ST- IND Z	
79	RDN	
80	ISG Y	
81	STO X	
82	ISG X	
83	GTO 05	
84	RCL Z	
85	RCL IND 00	Reduce b
86	LASTX	
87	*	
88	ST- IND Y	
89	RDN	
90	ISG X	
91	GTO 03	Reduce next row
92	GTO 01	Next pivot

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

	PREQ, NR	JM	LE
00	Counter	Counter	Counter
01	x_1, $-f_1$	x_1	b_1, x_1
02	x_2, $-f_2$	x_2	b_2, x_2
⋮	⋮	⋮	⋮
N	x_N, $-f_N$	x_N	b_N, x_N
N+1	f_{11}	f_{11}	a_{11}
N+2	f_{12}	f_{12}	a_{12}
⋮	⋮	⋮	⋮
$N+N^2$	f_{NN}	f_{NN}	a_{NN}
$N+N^2+1$	$-f_1$, x_1		
$N+N^2+2$	$-f_2$, x_2		
⋮	⋮		
$N+N^2+N$	$-f_N$, x_N		
*SIZE	N^2+2N+1	N^2+N+1	N^2+N+1

** PROGRAM BYTE REQUIREMENTS

LBL'PREQ		
LBL'NI		
LBL'CE		
LBL'OX		
END	448 BYTES	
LBL'NR		
END	99 BYTES	
LBL'JM		
END	112 BYTES	
LBL'N		
LBL'R		
LBL'W		
LBL'X		
LBL'Y		
LBL'Z		
END	112 BYTES	
LBL'LE		Complete pivoting
END	295 BYTES	
LBL'LE		Partial pivoting
END	192 BYTES	
LBL'LE		No pivoting
END	159 BYTES	

STATUS

SIZE * TOT. REG. ** USER MODE

ENG FIX SCI ON OFF

DEG RAD GRAD

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		JM executed	LE executed
01		DET=0	
02		I=IMAX	
03		E≤1.00E-6	
21			No printer
25		Size check	
29			Alpha use
55		Printer present	No printer

** PREQ BYTE REQUIREMENTS

LBL'PREQ	
LBL'NI	
LBL'CE	
LBL'OX	
LBL'NR	
LBL'JM	
LBL'LE	Complete pivoting
LBL'N	
LBL'R	
LBL'W	
LBL'X	
LBL'Y	
LBL'Z	
END	
1054 BYTES	151 registers

SYSTEM CONFIGURATION



012010

CARD LABELING



PROGRAMMED EQUATIONS
PREQ
PROGRAM REGISTERS NEEDED: 64

USERS' LIBRARY
PROGRAM NUMBER: 01201C

ROW 1 (1 : 3)

ROW 2 (4 : 13)

ROW 3 (13 : 20)

ROW 4 (21 : 23)

ROW 5 (23 : 29)

ROW 6 (29 : 35)

ROW 7 (35 : 40)

ROW 8 (40 : 44)

ROW 9 (45 : 48)

ROW 10 (48 : 53)

ROW 11 (54 : 60)

ROW 12 (61 : 66)

ROW 13 (67 : 70)

ROW 14 (70 : 75)

ROW 15 (76 : 83)

ROW 16 (84 : 94)

ROW 17 (95 : 101)

ROW 18 (102 : 110)

ROW 19 (111 : 118)

ROW 20 (119 : 124)

ROW 21 (124 : 130)

ROW 22 (131 : 140)

ROW 23 (141 : 151)

ROW 24 (151 : 156)

ROW 25 (157 : 162)

ROW 26 (163 : 169)

ROW 27 (170 : 176)

ROW 28 (177 : 182)

ROW 29 (183 : 190)

ROW 30 (190 : 195)

ROW 31 (196 : 200)

ROW 32 (201 : 206)

ROW 33 (207 : 211)

ROW 34 (212 : 216)

ROW 35 (217 : 219)

PROGRAMMED EQUATIONS
LE
PROGRAM REGISTERS NEEDED: 43

USERS' LIBRARY
PROGRAM NUMBER: 01201C

ROW 1 (1 : 5)

ROW 2 (6 : 12)

ROW 3 (13 : 22)

ROW 4 (22 : 30)

ROW 5 (30 : 39)

ROW 6 (40 : 48)

ROW 7 (49 : 56)

ROW 8 (57 : 65)

ROW 9 (65 : 73)

ROW 10 (74 : 81)

ROW 11 (82 : 88)

ROW 12 (89 : 97)

ROW 13 (98 : 106)

ROW 14 (106 : 113)

ROW 15 (114 : 121)

ROW 16 (122 : 131)

ROW 17 (132 : 140)

ROW 18 (140 : 147)

ROW 19 (148 : 157)

ROW 20 (157 : 164)

ROW 21 (165 : 172)

ROW 22 (173 : 181)

ROW 23 (182 : 186)

PROGRAMMED EQUATIONS
N
PROGRAM REGISTERS NEEDED: 16

USERS' LIBRARY
PROGRAM NUMBER: 01201C

ROW 1 (1 : 6)

ROW 2 (7 : 12)

ROW 3 (13 : 19)

ROW 4 (19 : 25)

ROW 5 (25 : 31)

ROW 6 (31 : 38)

ROW 7 (38 : 43)

ROW 8 (44 : 49)

ROW 9 (50 : 53)

PROGRAMMED EQUATIONS
PARTIAL PIVOTING LE
PROGRAM REGISTERS NEEDED: 28

USERS' LIBRARY
PROGRAM NUMBER: 01201C

ROW 1 (1 : 5)

ROW 2 (6 : 14)

ROW 3 (14 : 22)

ROW 4 (23 : 31)

ROW 5 (31 : 37)

ROW 6 (38 : 46)

ROW 7 (46 : 54)

ROW 8 (54 : 63)

ROW 9 (63 : 70)

ROW 10 (70 : 78)

ROW 11 (78 : 86)

ROW 12 (86 : 94)

ROW 13 (94 : 101)

ROW 14 (101 : 108)

ROW 15 (109 : 113)

PROGRAMMED EQUATIONS
NR
PROGRAM REGISTERS NEEDED: 15

USERS' LIBRARY
PROGRAM NUMBER: 01201C

ROW 1 (1 : 8)

ROW 2 (9 : 15)

ROW 3 (16 : 23)

ROW 4 (24 : 30)

ROW 5 (31 : 38)

ROW 6 (39 : 46)

ROW 7 (46 : 54)

ROW 8 (55 : 58)